

Instrucciones

(el lenguaje *assembly* RISC-V)

Arquitectura de Computadores

Todo computador debe tener instrucciones para ejecutar las operaciones aritméticas básicas; p.ej.:

add a,b,c —*suma las variables **b** y **c** y pone la suma en **a***

La notación (a este nivel) es rígida:

- cada instrucción aritmética ejecuta sólo una operación y siempre debe tener exactamente tres variables

P.ej., para sumar cuatro variables —**b**, **c**, **d** y **e**— y poner la suma en **a** se necesitan tres instrucciones, cada una en una línea por sí sola:

```
add a,b,c  
add a,a,d  
add a,a,e
```

P.ej., ¿cómo traduciría el compilador el siguiente segmento de un programa en C o Python a instrucciones del lenguaje assembly?

```
a ← b + c  
d ← a - e
```

Respuesta:

```
add a,b,c  
sub d,a,e
```

P.ej., ¿qué haría el compilador en el caso de la siguiente sentencia?

$$f \leftarrow (g+h) - (i+j)$$

Respuesta:

```
add t0,g,h  
add t1,i,j  
sub f,t0,t1
```

t0 y **t1** son variables auxiliares para almacenar temporalmente los resultados de **g+h** e **i+j** antes de hacer la resta

A diferencia de un programa en C o Java, los tres operandos de las instrucciones aritméticas deben corresponder a **registros**:

- hay un número limitado de registros, p.ej., 32
- en RISC-V, son de 64 bits de largo c/u

Para los nombres de los registros, RISC-V usa una **x** seguida del número del registro (un número entre 0 y 31)

P.ej., si las variables **f**, **g**, **h**, **i** y **j** están asignadas a los registros **x19**, **x20**, **x21**, **x22** y **x23**

... y usamos los registros **x5** y **x6** para almacenar los resultados intermedios

... entonces el código compilado para **$f \leftarrow (g + h) - (i + j)$** sería

```
add x5,x20,x21
add x6,x22,x23
sub x19,x5,x6
```

Los lenguajes de programación también tienen arreglos (y otras estructuras), que pueden contener muchos más datos que la cantidad de registros del computador

Estas estructuras, que simplemente no caben en 32 registros, se almacenan en la memoria del computador:

- que tiene capacidad para almacenar miles de millones de datos

... → el lenguaje assembly debe tener instrucciones para *transferir datos entre la memoria y los registros*: **ld** y **sd**

La memoria es como un arreglo unidimensional muy grande de celdas o casillas, en que la *dirección* (de memoria) actúa como índice:

- **ld** y **sd** deben especificar la dirección de la celda involucrada
- **ld** y **sd** especifican la dirección de modo *indirecto*, por una constante y un registro → la suma de la constante y el contenido del registro

procesador


ALU
+
32 registros

memoria



celdas
(miles de
millones)

bus

memoria




valor usado
como ejemplo $2^{32}-1$



direcciones

= números correlativos que
corresponden a las posiciones
relativas de los bytes

bit: 0 o 1



byte = 8 bits consecutivos

ld : copia datos desde la memoria a un registro (*load doubleword*)

P.ej., ¿cómo traduce el compilador la sentencia **$g \leftarrow h + A[8]$** en C?

- el compilador coloca el arreglo **A** en memoria \rightarrow sabe cuál es la dirección de memoria de **A[0]** (el primer elemento de A) y coloca esta dirección en un registro, llamado el *registro base*; supongamos que el registro base es **x22**
- el compilador asocia variables con registros \rightarrow supongamos que las variables **g** y **h** están asociadas a los registros **x20** y **x21**

Entonces **$g \leftarrow h + A[8]$** es traducida por el compilador así:


```
ld  x9,8(x22)
add x20,x21,x9
```

—*dos instrucciones en assembly:*

—... *primero, copia A[8] al registro x9*

—... *luego, suma y pone el resultado en g*

nombrados de los
registros en RISC-V



La memoria, en la práctica en todos los computadores, está organizada en *bytes* (8 bits consecutivos)

... en que cada byte tiene una dirección: 0, 1, 2,


Los computadores modernos, sin embargo, operan sobre *palabras* (*words*) de 4 bytes consecutivos (32 bits),

... o, como en el caso de RISC-V, incluso sobre *palabras dobles* (*double-words*) de 8 bytes consecutivos (64 bits)

... por lo que las direcciones de estas palabras dobles son 0, 8, 16, ...

- ... es decir, la dirección correspondiente al primer byte de la palabra
- ... de modo que el ejemplo anterior requiere un ajuste

Veámoslo con la siguiente instrucción



palabra doble
= 8 bytes consecutivos (64 bits)

memoria RISC-V



direcciones (de las palabras dobles): corresponden
a la dirección de uno de los 8 bytes de la palabra
—el byte con la dirección numéricamente menor

sd : copia datos desde un registro a la memoria (*store doubleword*):

$A[12] \leftarrow h + A[8]$ —*sentencia en C, Java, etc.*

De nuevo, **x22** contiene la dirección de **A[0]** y **x21** está asociado a **h**:

—*tres instrucciones en assembly:*

ld x9, 64(x22)

add x9, x21, x9

sd x9, 96(x22)

—... *copia (load) A[8] al registro x9*

—... *suma h a A[8]*

—... *copia (store) el registro x9 a A[12]*

En muchas operaciones, aparecen operandos constantes, o *inmediatos*: el valor numérico propiamente tal

RISC-V tiene versiones de las instrucciones aritméticas en que uno de los operandos es una constante

P.ej., la instrucción *add-immediate*, or **addi**:

`addi x22,x22,4` —corresponde a la sentencia $x22 \leftarrow x22 + 4$

Es útil poder operar sobre conjuntos de bits dentro de una palabra o incluso sobre bits individuales —operaciones lógicas, típicamente:

AND, OR, XOR, NOT, *shift left*, *shift right* y *shift right arithmetic*

Los *shifts* mueven todos los bits de una palabra a la izquierda o a la derecha, llenando los bits que quedan “vacíos” con 0s:

P.ej., si **x19** contiene

0000 0000 0000 0000 0000 0000 0000 1001₂ (= 9₁₀)

... y ejecutamos la instrucción

`slli x11,x19,4` —*shift left logical immediate*

... → *shift left* de 4 posiciones y dejamos el resultado en **x11**:

0000 0000 0000 0000 0000 0000 1001 0000₂ (= 144₁₀)

El lenguaje assembly también tiene la instrucción *shift right logical immediate* (**srli**)

... y las versiones de estas instrucciones en que la cantidad de bits desplazados es especificada en un registro y no como una constante:

- *shift left logical* (**sll**) y *shift right logical* (**srl**)

En el caso de *shift right*, el lenguaje tiene además las instrucciones en que los bits que quedan “vacíos” a la izquierda se llenan con copias del bit de signo original (y no necesariamente con 0s):

- *shift right arithmetic immediate* (**srai**) y *shift right arithmetic* (**sra**)

and y **or** : operaciones bit a bit entre los contenidos de dos registros

P.ej., si

x10 = 0000 0000 0000 0000 0011 1100 0000 0000₂

x11 = 0000 0000 0000 0000 0000 1101 1100 0000₂

... entonces la ejecución de

and x9,x10,x11 —*la sentencia en C podría ser* $r \leftarrow s \ \& \ t$

... deja

x9 = 0000 0000 0000 0000 0000 1100 0000 0000₂

Todo lenguaje de programación

... incluyendo los lenguajes de máquina y los lenguajes *assembly*

... deben tener instrucciones para controlar la ejecución condicional de otras instrucciones:

- dependiendo de los datos de entrada y de los valores creados durante la ejecución del programa,
- ... se ejecutan unas instrucciones u otras

En el assembly RISC-V:

`beq registro1, registro2, L1`

... significa “ir a la instrucción etiquetada *L1* si el valor en *registro1* **es igual** al valor en *registro2*” —*branch if equal*

`bne registro1, registro2, L1`

... significa “ir a la instrucción etiquetada *L1* si el valor en *registro1* **no es igual** al valor en *registro2*” —*branch if not equal*

P.ej., compilemos la siguiente sentencia a nuestro assembly:

```
if (i = j):  
    f ← g + h  
else:  
    f ← g - h
```

Si las variables **f**, **g**, **h**, **i** y **j** corresponden a los registros **x19** a **x23**:

```
        bne x22,x23,Else  
        add x19,x20,x21  
        beq x0,x0,Exit  
Else:   sub x19,x20,x21  
Exit:
```

beq x0,x0,Exit es en la práctica un *branch incondicional* (ya que **x0** es siempre igual a **x0**) a la instrucción etiquetada **Exit**

El *assembler* (ensamblador) libera al compilador y al programador de tener que calcular las direcciones (de memoria) para los *branches*

Las decisiones también son importantes para repetir la ejecución de una computación —*loops*

Las mismas instrucciones condicionales sirven para estos casos

P.ej., compilemos el siguiente loop:

```
while (save[i] = k):  
    i ← i+1
```

Supongamos que **i** y **k** corresponden a los registros **x22** y **x24**, **save[0]** está en **x25**, y queremos guardar **save[i]** en **x9**

Primero, necesitamos la dirección de **save[i]**: sumamos **i** a la base de **save**, por lo que antes hay que multiplicar **i** por **8**, lo que en este caso hacemos empleando *shift left logical immediate*:

Loop:	<code>slli</code>	<code>x10,x22,3</code>	—multiplicación i*8
	<code>add</code>	<code>x10,x10,x25</code>	—suma i + save[0]
	<code>ld</code>	<code>x9,0(x10)</code>	—carga save[i] en x9
	<code>bne</code>	<code>x9,x24,Exit</code>	—el test del loop
	<code>addi</code>	<code>x22,x22,1</code>	—incremento de i
	<code>beq</code>	<code>x0,x0,Loop</code>	—volvemos al principio

Exit:

Además de chequear igualdad (**beq**) o desigualdad (**bne**), a veces es útil chequear si una variable es menor que otra

La instrucción

blt *registro1, registro2, L1*

... salta a la instrucción etiquetada *L1* si *registro1* < *registro2*, cuando los valores son tratados como números con signo:

- **bltu** salta si *registro1* < *registro2*, cuando los valores son tratados como números sin signo (*branch if less than, unsigned*)

También tenemos las instrucciones **bge** y **bgeu**:

- saltan si el valor en el primer registro es al menos tan grande como el valor en el segundo registro (*branch if greater than or equal*)
... para números con signo y sin signo, respectivamente

Una *función* es una herramienta que los programadores usan para estructurar un programa:

- los programas son más fáciles de entender
- el código puede ser reusado

Los *parámetros* de la función actúan como una interfaz entre la función y el resto del programa y datos:

- pasan valores
- devuelven resultados

Al ejecutar la función, el programa debe seguir estos pasos:

- poner los parámetros en un lugar donde la función tenga acceso a ellos
- transferir el control a la función
- obtener la memoria necesaria para la ejecución de la función
- ejecutar la tarea correspondiente
- poner el resultado en un lugar donde el programa que hizo la llamada tenga acceso a él
- devolver el control al punto de origen, ya que una función puede ser llamada desde varios puntos en un programa

Los registros son el lugar de más rápido acceso para tener datos en un computador

En RISC-V tenemos 32 registros; los usamos así:

- **x10 – x17**: ocho registros para pasar parámetros o retornar valores
- **x1**: un registro para la dirección de retorno para volver al punto de origen

RISC-V tiene además la instrucción *jump and link* solo para funciones:

- **jal x1, direccióndeLaFunción**

... salta a la dirección *direccióndeLaFunción*

... y simultáneamente guarda la dirección de la siguiente instrucción —la **dirección de retorno**— en el registro **x1**

(la *dirección de retorno* es necesaria porque la misma función podría ser llamada desde varias partes del programa)

... y la instrucción *jump and link register* para volver desde una función:

- **jalr x0,0(x1)**
- ... salta a la dirección almacenada en x1 (x0 está *hard-wired* con el valor 0)

El programa que hace la llamada primero pone los valores de los parámetros en los registros **x10 – x17**

... y luego ejecuta **jal x1, F** para saltar (al código correspondiente) a la función **F**:

- por supuesto, el registro **PC** —*program counter*— guarda la dirección de la instrucción que está siendo ejecutada en este momento

... por lo que **jal** pone realmente **PC + 4** en **x1**

La función **F** entonces hace sus cálculos (o lo que sea)

... pone los resultados en los mismos registros **x10 – x17**

... y finalmente devuelve el control al programa que hizo la llamada ejecutando **jalr x0, 0(x1)**

Si al llamar a una función el compilador necesita más registros que los ocho registros **x10 – x17**, entonces

... todos los registros adicionales que sean usados deben ser finalmente restaurados a los valores que tenían antes de que se produjera la llamada —recuperar sus valores originales:

- → los valores originales de esos registros deben ser guardados en la memoria mientras dura la llamada a la función

La estructura de datos para esto es un ***stack*** —una cola del tipo “el último en entrar es el primero en salir”, o LIFO

... y un puntero —el *stack pointer* o **sp**, almacenado en el registro **x2**— a la dirección más recientemente reservada en el stack para estos registros adicionales

P.ej., ¿cuál es el código assembly compilado de la siguiente función?

```
int xmpl(int g, int h, int i, int j):  
    int f  
     $f \leftarrow (g+h) - (i+j)$     —ver diap. #4  
    return f
```

Los parámetros **g**, **h**, **i** y **j** corresponden, como dijimos, a los registros **x10**, **x11**, **x12** y **x13**

La variable **f** corresponde al registro **x20** (este es un registro adicional)

Y el código compilado comienza con la etiqueta **xmpl**:

Además, para calcular el valor de **f**, usamos dos registros temporales, **x5** y **x6** (también adicionales)

Así, lo primero es guardar en el stack los valores de los registros **\$s0**, **\$t0** y **\$t1** —tres palabras dobles (24 bytes):

```
addi    sp, sp, -24
sd      x5, 16(sp)
sd      x6, 8(sp)
sd      x20, 0(sp)
```

Luego, hay que calcular el valor de la expresión y guardarlo en **\$s0**:

```
add x5, x10, x11
add x6, x12, x13
sub x20, x5, x6
```

... y hay que retornar este valor, guardándolo en **x10**:

```
addi x10, x20, 0
```

Antes de volver al punto donde se produjo la llamada, restauramos los valores de los tres registros guardados en el stack:

```
ld    x20, 0(sp)
ld    x6, 8(sp)
ld    x5, 16(sp)
addi  sp, sp, 24
```

... y finalmente retornamos, ejecutando

```
jalr  x0, 0(x1)
```


En realidad, según lo que vimos antes (diap. #5), no es necesario guardar en el stack los valores de los registros **x5** y **x6** —no representan valores de variables del programa, sino resultados intermedios:

- en RISC-V, los registros **x5** a **x7** y **x28** a **x31** son registros temporales que no son preservados durante una llamada a una función
- ... mientras que los registros **x8**, **x9**, y **x18** a **x27** son registros que, si son usados, deben ser preservados durante una llamada a una función

Esto significa que en la práctica podemos ahorrarnos los dos *loads* y los dos *stores* de **x5** y **x6**

¿Qué pasa en el siguiente caso?

```
int factorial(int n):  
    if n < 1:  
        return 1  
    else:  
        return n*factorial(n-1)
```

Como la función es recursiva, ahora también es necesario guardar en el stack los valores de los registros **\$a0** y **\$ra**, para poder recuperarlos al volver a la llamada anterior:

```
factorial:  
    addi    $sp, $sp, -8  
    sw      $ra, 4($sp)  
    sw      $a0, 0($sp)
```

Luego, comparamos **n** con 1, y vamos al *label* **L1** si $n \geq 1$:

```
slti    $t0, $a0, 1  
beq $t0, $zero, L1
```

Si $n < 1$, entonces **factorial** devuelve 1, a través del registro **\$v0**:

```
addi    $v0, $zero, 1  
addi    $sp, $sp, 8  
jr      $ra
```

Si $n \geq 1$, **n** es decrementado y **factorial** es llamado recursivamente:

L1:

```
addi    $a0, $a0, -1  
jal factorial
```

De ahí, empezamos a “salir” de (esta llamada de) la función:

```
lw    $a0, 0($sp)
lw    $ra, 4($sp)
addi  $sp, $sp, 8
```

... ponemos en **\$v0** el producto de **\$a0** y **\$v0**:

```
mul $v0, $a0, $v0
```

... y finalmente, **factorial** salta de nuevo a la dirección de retorno:

```
jr    $ra
```

Para la mayoría de las funciones en la práctica, es suficiente con

... 4 parámetros (los registros 4 al 7)

... 2 registros para el valor de retorno (los registros 2 y 3)

... y otros 10 registros temporales (los registros 8 al 15, y los registros 24 y 25)

El stack se usa también para almacenar variables locales de la función que no caben en los registros, p.ej., arreglos

La porción del stack que contiene los registros guardados y las variables locales de una función se conoce como **marco de activación**

Un (nuevo) registro —***frame pointer*** (**\$fp**, o el registro 30, en MIPS)— apunta a la primera palabra del marco de una función (normalmente, un registro con el valor de un parámetro):

- el *stack pointer* puede cambiar durante la ejecución de una función → las referencias a una variable local podrían tener diferentes *offsets*, dependiendo de dónde están en el marco
- el *frame pointer* tiene un valor fijo, conveniente para las referencias a memoria locales
- al hacerse una llamada, el *frame pointer* recibe el contenido del *stack pointer*

... y al volver de la llamada, el *stack pointer* es actualizado con el valor del *frame pointer*

Espacio para otros datos en el stack

Espacio para otros datos en el heap

Las instrucciones se almacenan en el computador representadas como números binarios:

- cada parte de una instrucción —p.ej., el operador y cada uno de los tres operandos— puede ser considerada como un número individual
- ... y al poner estos números juntos, siguiendo el formato de la instrucción, forman la instrucción

Como muchos operandos son registros, necesitamos una convención para asociar los nombres de los registros con números:

- p.ej., los registros \$s0 a \$s7 corresponden a los números 16 a 23
- ... los registros \$t0 a \$t7 corresponden a los números 8 a 15